

EEL7030 Microprocessadores – Laboratório 7

Prof. Raimes Moraes

Interface Serial do 8051

Compile e execute o programa abaixo que transmite, serialmente, a cadeia de 16 caracteres: 'Microcontrolador'. A transmissão é assíncrona sem envio de bit de paridade à taxa de 9600 bauds (bit/s). Para visualizar dados transmitidos, selecionar no menu: *View, Serial Windows, UART #1*. OBS: Informar cristal a ser utilizado no hardware (11,059 MHz) para o Keil: *Project - Options for Target* (ou Alt + F7); modificar o campo Xtal

```
RESET      EQU    00H
LTSERIAL    EQU    23H ; local tratador
STATE       EQU    20H

                ORG RESET    ;PC=0 depois de reset
                JMP INICIO

                ORG    LTSERIAL
                CLR     TI
                MOV     STATE,#1H
                RETI

INICIO:        MOV     IE,#10010000B
                MOV     SCON,#01000000B
                MOV     TMOD,#00100000B
                MOV     TH1,#0FDH
                MOV     TL1,#0FDH
                MOV     PCON,#0H
                SETB    TR1

                MOV     STATE,#0H
                MOV     R0,#STATE
                MOV     DPTR,#TABELA
                MOV     R1,#1
                MOV     SBUF,#'M'

VOLTA:         CJNE    @R0,#1,VOLTA
                MOV     STATE,#0H
                MOV     A,R1
                MOVC    A,@A+DPTR
                MOV     SBUF,A
                INC     R1
                CJNE    R1,#16,VOLTA
                CLR     TR1
                JMP     $

TABELA:        DB 'Microcontrolador'
                END
```

Exercícios:

- 1) Modifique o programa acima para que o mesmo apenas receba dados e os armazene na memória interna entre os endereços 30h e 35h. Armazenar os dados de forma cíclica; ou seja, ao término do preenchimento de i:0x35, voltar a preencher a partir de i:0x30. Para enviar dado para a recepção serial pela *Debug session*, tem-se que digitar (por exemplo) SIN=0x33 na janela de comando do Keil (no *prompt (>)* da janela de comando).
- 2) Programar a interface serial do 8051 para enviar os caracteres de 41h a 61h de forma cíclica, utilizando a taxa de 62,5 kbits/s. Transmitir 8 bits de dados mais bit de paridade.
- 3) Modificar programa anterior para que o mesmo também receba dados e os armazene na memória interna entre os endereços 30h e 35h. Armazenar os dados de forma cíclica.
- 4) Fazer um programa para transmitir pela interface serial os valores de 41H a 61H de forma cíclica. Devem ser transmitidos 8 bits de dados à taxa de 19,2 kbits por segundo. O programa deve também receber dados, armazenando os valores recebidos na memória externa entre os endereços 0h e 0009h. Após completar o preenchimento desta região de memória, voltar a preenchê-la com os dados recebidos a partir do endereço inicial.